

Raccolta delle Procedure di Analisi Territoriale

Markus M. Hedorfer

Istituto Universitario di Architettura di Venezia — Dipartimento di Urbanistica

Santa Croce 1957, I-30135 Venezia VE

Telefono: ++39-041-2572137

Posta elettronica: hedorfer@iuav.it

WWW home page: <http://cidoc.iuav.it/~hedorfer>

Sito FTP: <ftp://cidoc.iuav.it/pub/hedorfer>

29 giugno 1999

Indice

Elenco delle tabelle	1
Elenco delle figure	2
1 Simulazione delle Superfici Urbane Contigue (CUS)	3
Principio	3
Implementazione	4
Procedura	4
2 Modello Urbano di Distribuzione della Popolazione (UPDM)	5
Principio	6
Implementazione	7
Procedura	9
3 Modello di Integrazione ‘Fuzzy’ dell’Uso del Suolo (FLCIM)	11
Principio	12
Implementazione	12
Procedura	16

Elenco delle figure

Elenco delle tabelle

1	Dati territoriali di base richiesti per la simulazione delle superfici urbane contigue (CUS)	4
2	Dati territoriali di base richiesti per il modello urbano di distribuzione della popolazione (UPDM)	6
3	Dati territoriali di base richiesti per il modello di integrazione 'fuzzy' dell'uso del suolo (FLCIM)	13

Elenco delle figure

1	Schema procedurale delle operazioni di analisi spaziale del FLCIM	14
---	---	----

1 Simulazione delle Superfici Urbane Contigue (CUS)

La mancanza, in molte situazioni, di un modello numerico degli usi del suolo a scala urbana da un lato, ma anche particolari esigenze analitiche di tipo quantitativo, rendono necessaria una procedura che individui in modo dinamico le parti del territorio in esame definibili come superfici urbane o urbanizzate. Tra le esigenze analitiche si possono citare

- la generalizzazione di dati geospaziali per renderli compatibili con altri dati con errori posizionali maggiori,
- l'individuazione di aggregati urbani in base a diverse ipotesi deterministiche,
- il calcolo del potenziale residenziale o insediativo in generale oppure
- il calcolo del tasso di antropizzazione del territorio.

Principio

La simulazione delle superfici urbane contigue (CUS) si basa sul principio di base che ogni opera edilizia esercita un'influenza sul territorio che è maggiore della sola superficie effettivamente coperta e che, quindi, investe un determinato intorno ad esso che può essere funzionalmente assoggettato alla stessa opera edilizia. Considerando, per esempio un edificio residenziale isolato — e cioè non inserito in alcun contesto urbano — il terreno antropizzato è probabilmente costituito, oltre alla superficie occupata dall'edificio, anche dalle aree immediatamente circostanti e presumibilmente destinata a giardino o ad altra funzione di pertinenza della casa.

La tecnica analitica geospaziale che viene impiegata nella simulazione delle CUS è quella del *buffering*, vale a dire della costruzione di una fascia a profondità costante attorno all'edificio stesso che viene unita alla superficie occupata dall'edificio e alle fasce attorno ad edifici eventualmente ubicati nelle vicinanze.

La distanza da applicare durante la costruzione di queste fasce di influenza dipende fortemente dal proposito analitico posto. Alcune situazioni — come per esempio l'individuazione delle aree urbanizzate in mancanza di fonti di dati più attendibili — potrebbero impiegare un valore costante (per esempio 40 m), mentre altre potrebbero applicare degli algoritmi più complessi come per esempio

$$distanza = \frac{distanza_base}{2} \left[(1 + vulnerabilità)^2 + \sqrt{\frac{altezza}{3\text{ m}}} \right]$$

dove *distanza_base* potrebbe essere la distanza da applicare nel caso di edifici con *altezza* = 3 m e *vulnerabilità* = 0.

1 Simulazione delle Superfici Urbane Contigue (CUS)

Tabella 1: Dati territoriali di base richiesti per la simulazione delle superfici urbane contigue (CUS)

Dato	Entità	Note
Edifici residenziali	punti	richiesto
Edifici residenziali	poligoni	consigliato

Implementazione

Per poter eseguire la simulazione delle CUS è necessario disporre almeno di una copertura puntuale in cui ogni punto rappresenta un edificio attorno al quale creare una fascia di influenza. Tuttavia considerando che le dimensioni degli edifici sono di norma sensibilmente differenti, è consigliato utilizzare coperture poligonali, dove gli edifici vengono rappresentati con le loro estensioni reali. Le operazioni da eseguire sono quindi le seguenti.

1. *Calcolo delle profondità individuali* delle fasce di influenza. Per agevolare l'operazione di bufferizzazione, occorre creare una copia della copertura di partenza e aggiungere nella tabella degli attributi poligonali o puntuali un nuovo campo numerico (per questioni di compatibilità, tale campo dovrebbe essere sufficientemente capiente per contenere sia numeri interi che a virgola mobile) per accogliere la profondità della fascia attorno ogni edificio. Nel caso di profondità costanti, tale valore sarà uguale per ogni record della tabella, nel caso invece di profondità variabili ogni record potrà riportare un valore diverso.
2. *Creazione delle fasce di influenza*. Viene utilizzata la procedura **BUFFER** del software GIS specificando una tolleranza adeguata a dati territoriali tecnici di tipo urbano o metropolitano (50 cm).

L'aggiunta del nuovo campo numerico e il calcolo delle profondità individuali avvengono utilizzando una copia della copertura di partenza che, al termine delle operazioni, viene rimossa dalla base di dati.

Procedura

La procedura riportata qui di seguito rappresenta un'implementazione completa della simulazione di CUS per Arc/Info sotto forma di un macro-comando con la sintassi

```
CUS <in_cover> <out_cover> <value | expression>  
dove
```

in_cover indica la copertura puntuale o poligonale degli edifici,

out_cover la copertura poligonale delle fasce di influenza da creare,

value un valore numerico nell'ipotesi di fasce a profondità costante e

expression un'espressione di selezione riferita alla tabella degli attributi poligonali <icov>.PAT nell'ipotesi di fasce a profondità variabile.

```

&args icov ocov expr:rest

/* Parameter tests
&if [null %expr%] &then
    &return Usage: CUS <in_cover> <out_cover> <value | expression>
&sv ico = [upcase %ico%]
&sv oco = [upcase %oco%]
&sv icoen = [entryname %ico%]
&sv ocoen = [entryname %oco%]

/* Process point or polygon features (or exit)
&if [exists %ico% -polygon] &then
    &sv fea = poly
&else
    &do
        &if [exists %ico% -point] &then
            &sv fea = point
        &else
            &return Warning: Coverage %ico% must have either point or polygon topology.
    &end

/* Create temporary coverage and calculate buffer distance
copy %ico% xtemp
&sv ows = [show workspace]
workspace [dir %icov%]
tables
additem %icoven%.PAT dist 8 18 f 5
select %icoven%.PAT
calculate dist = %expr%
quit
workspace %ows%

/* Create buffer and remove temporary coverage
buffer xtemp %ocov% dist # # 0.5 %fea%
kill xtemp all

```

2 Modello Urbano di Distribuzione della Popolazione (UPDM)

Nell'ambito di molte operazioni di analisi territoriale a scala urbana che utilizzano dati censuari — soprattutto quelle funzionali alla pianificazione urbanistica — le sole informazioni a livello di sezioni di censimento non sono sempre sufficienti al fine di ottenere indicazioni utili sulla distribuzione territoriale della popolazione insediata o anche di altri insediamenti a carattere non residenziale. Contemporaneamente le basi di dati anagrafiche, che permetterebbero una più accurata localizzazione dei fenomeni insediativi, nella maggior parte dei ca-

Tabella 2: Dati territoriali di base richiesti per il modello urbano di distribuzione della popolazione (UPDM)

Dato	Entità	Attributi	Note
Sezioni di censimento	poligoni	codici delle sezioni	richiesto
Edifici residenziali	punti		richiesto
Edifici residenziali	poligoni		consigliato
Edifici residenziali	poligoni	altezze degli edifici	opzionale
Edifici residenziali	poligoni	volumi degli edifici	opzionale

si non sono sufficientemente interfacciate a sistemi informativi geospaziali da permettere un loro utilizzo immediato nell'ambito delle analisi spaziali.

Principio

Il modello urbano di distribuzione della popolazione (UDPM) predispone una semplice ripartizione della popolazione all'interno delle singole sezioni di censimento utilizzando il dato di base degli edifici residenziali che di norma è facilmente reperibile dalle comuni opere cartografiche numeriche di base (in Italia solitamente la «Carta Tecnica Regionale Numerica»).

A seconda della disponibilità di dati attribuiti più o meno dettagliati sugli edifici residenziali sarà possibile effettuare delle interpolazioni più o meno attendibili. Nell'ipotesi che l'unico dato attribuito degli edifici residenziali sia quello localizzativo (entità puntuali in rappresentanza delle superfici coperte), a ciascun edificio viene attribuito una parte eguale della popolazione complessiva della sezione di censimento entro la quale ricade:

$$p_{ij} = \frac{p_i}{n_i} \quad (1)$$

dove p_{ij} indica la popolazione del j -esimo edificio dell' i -esima sezione di censimento, p_i la popolazione dell' i -esima sezione di censimento e n_i il numero di edifici compresi entro l' i -esima sezione di censimento. Nel caso, invece, si disponga degli edifici residenziali sotto forma di entità poligonali — e quindi conoscendo l'estensione della superficie coperta di ciascun edificio — tale metodo di ripartizione potrà essere migliorato ponderando la quota sulla popolazione complessiva di ciascuna sezione di censimento in base alla superficie occupata dagli edifici applicando la formula

$$p_{ij} = \frac{A_{ij}p_i}{\sum_{j=1} A_{ij}} \quad (2)$$

dove A_{ij} indica la superficie coperta di un singolo edificio residenziale ij -esimo. Risultati ancora più vicini alla realtà vengono ovviamente ottenuti se, oltre alla superficie coperta, si conosce anche l'altezza di ogni singolo edificio permettendo

così di ottenere un'indicazione sul volume spesso sufficiente ai fini delle analisi territoriali urbane. La formula da applicare diventa di conseguenza

$$p_{ij} = \frac{A_{ij}h_{ij}p_i}{\sum_{j=1}^{n_i} A_{ij}h_{ij}} \quad (3)$$

oppure

$$p_{ij} = \frac{V_{ij}p_i}{\sum_{j=1}^{n_i} V_{ij}} \quad (4)$$

se si dispone direttamente del dato volumetrico corretto per ogni edificio (h_{ij} indica l'altezza media di un edificio ij -esimo, mentre V_{ij} il suo volume).

Implementazione

Si suppone di disporre, nell'ambito di un sistema informativo geospaziale a carattere urbano o metropolitano, dei seguenti dati territoriali di base.

1. Copertura poligonale **CENSUS** che contiene, per tutte le sezioni di censimento interessate dal GIS, nella propria tabella degli attributi poligonali l'identificatore numerico univoco della sezione di censimento **CTR-ID** e l'ammontare complessivo della popolazione residente **PR**¹ corrispondente alla variabile p_i discussa sopra.
2. Copertura puntuale **HOUSESP** che contiene un insieme di punti in rappresentanza degli edifici residenziali relativi al territorio in questione.
3. Copertura poligonale **HOUSESA** in alternativa alla copertura puntuale **HOUSESP** che contiene le estensioni superficiali (le superfici coperte) degli edifici residenziali e, nella propria tabella degli attributi poligonali, un campo da impiegare per la ponderazione dei pesi dei singoli edifici. Solitamente, tale campo è il campo metrico **AREA** che quantifica la superficie dei poligoni espressa nel quadrato dell'unità di misura della copertura (m^2 nel caso di m). Altri campi potrebbero essere **H** (altezza media dell'edificio) o direttamente **V** (volume).

I singoli passaggi da eseguire saranno quindi i seguenti.

1. *Estrazione della copertura puntuale HOUSESP* nel caso si disponga della copertura poligonale **HOUSESA** al fine di poter eseguire correttamente il passaggio successivo. Il metodo più indicato è l'estrazione dalla copertura poligonale delle etichette dei poligoni mediante selezione, la loro copiatura verso una nuova copertura e quindi la costruzione di una topologia

¹Variabile d'esempio. Ovviamente, il modello può essere applicato a qualsiasi variabile dei dati censuari. Nel caso la variabile di riferimento non riguardi la popolazione insediata, occorrerà sostituire il dato degli edifici residenziali con il dato appropriato.

adimensionale (puntuale) della nuova copertura. Se il campo di ponderazione utilizzato è AREA, sarà inoltre necessario rinominare tale campo (per esempio in SC per «superficie coperta») per evitare che i valori archiviati vengano alterati dalla procedura di costruzione dei dati topologici.

2. *Sovrapposizione «point-in-polygon»* per assegnare ciascun edificio residenziale ad una sezione di censimento. Eseguendo il comando


```
identity HOUSESP CENSUS HOUSESO point 0.5 join
```

 viene creata una nuova copertura puntuale HOUSESO identica a quella iniziale HOUSESP eccetto per il fatto che a ciascun edificio residenziale saranno assegnate tutte le variabili già contenute nella tabella degli attributi poligonali delle sezioni di censimento. Il valore di 0,5 del quinto parametro indica la tolleranza, espressa qui in metri, che si assume per un dato territoriale tecnico costruito per essere rappresentato in scala 1:5.000. La parola chiave JOIN indica invece che tutti i campi della tabella degli attributi poligonali CENSUS.PAT dovranno essere ricopiati nella tabella HOUSESO.PAT.
3. *Calcolo del numero di edifici e della sommatoria dei valori di ponderazione* per ogni sezione di censimento. Eseguendo il comando


```
frequency HOUSESO.PAT CENSUS.RHO
```

 e utilizzando — nel successivo dialogo interattivo — il campo CTR-ID come raggruppatore e il campo di ponderazione (N in assenza di ponderazione, SC nel caso del valore delle superfici coperte e V nel caso dei volumi reali degli edifici) come campo del quale calcolare le somme, si otterrà la tabella risultato CENSUS.RHO che conterrà i campi CTR-ID, N e il campo di ponderazione se questo è diverso da N.
4. *Calcolo della popolazione attribuita ad ogni singolo edificio.* Impostando, a partire dalla tabella HOUSESO.PAT, due relazioni rispettivamente verso la tabella degli attributi poligonali delle sezioni di censimento CENSUS.PAT e la tabella delle sommatorie CENSUS.RHO, la popolazione attribuita ai singoli edifici viene calcolata eseguendo l'arrotondamento (per ottenere valori interi) del rapporto fra il prodotto del valore di ponderazione del singolo edificio (contenuta nella tabella HOUSESO.PAT) e la popolazione complessiva della sezione di censimento (nella tabella CENSUS.PAT) e la sommatoria dei valori di ponderazione (nella tabella CENSUS.RHO)

$$PR = \text{int}((W * \text{CENSUS.PAT} // PR) / \text{CENSUS.RHO} // W) + 0.5)$$
 dove W indica il rispettivo campo dal quale recuperare i valori di ponderazione e il simbolo // il separatore tra i nomi della tabella relazionata e il campo.
5. *Ricomposizione della copertura poligonale* degli edifici residenziali. Nel caso la copertura iniziale fosse di tipo poligonale, l'ultimo passaggio consisterà nella copiatura degli elementi lineari dalla copertura iniziale HOUSESA verso quella prodotta HOUSESO e nella ricostruzione delle informazioni topologiche di nodi, linee e poligoni. Il risultato finale sarà quindi una copertura poligonale identica a quella di partenza che è stata arricchita, per ogni edificio residenziale, del numero di identificazione della sezione di censimento a cui appartiene e del numero (intero) di abitanti assegnato

in base all'UPDM. Occorre sottolineare che il procedimento non altera la struttura poligonale essendo stato esplicitamente evitata un'operazione di sovrapposizione «*polygon-in-polygon*».

Procedura

La procedura riportata qui di seguito rappresenta un'implementazione completa dell'UPDM per Arc/Info sotto forma di un macro-comando con la sintassi

```
UDPM <census> <houses_point> <houses_out> {houses_poly} {weight_item}
dove
census indica la copertura poligonale delle sezioni di censimento,
houses_point la copertura degli edifici residenziali rappresentati con elementi
puntuali,
houses_out la copertura puntuale o poligonale finale degli edifici residenziali,
houses_poly l'eventuale copertura degli edifici residenziali di partenza rappre-
sentati con elementi poligonali e
weight_item l'eventuale campo contenente i valori da utilizzare per la ponde-
razione dei pesi dei singoli edifici.
```

```
&args cen rhp rho rha wit

/* Parameter tests
&if [null %rho%] &then
  &do
    &type Usage: UPDM <census> <houses_point> <houses_out>
    &type {houses_poly} {weight_item}
    &return
  &end
&sv cen = [upcase %cen%]
&sv rhp = [upcase %rhp%]
&sv rho = [upcase %rho%]
&sv cenen = [entryname %cen%]
&sv rhpen = [entryname %rhp%]
&sv rhoen = [entryname %rho%]
&if not [exists %cen% -polygon] &then
  &do
    &if not [exists %cen% -cover] &then
      &return Warning: Census coverage %cen% not found.
    &else
      &return Warning: Census coverage %cen% has no polygon topology.
    &end
&if [null %wit%] &then
  &do
    &sv wit = 1
    &sv wis = N
  &end
&else
  &do
```

2 Modello Urbano di Distribuzione della Popolazione (UPDM)

```
&sv wit = [upcase %wit%]
&if %wit% = AREA &then
    &sv wit = SC
&sv wis = %wit%
&if [null %rha%] &then
    &sv rhx = %rhp%
&else
    &sv rhx = %rha%
&if not [iteminfo %rhx%.PAT -info %wit% -exists] &then
    &return Warning: Weight item %wit% does not exist within %rhx%.PAT
&sv wit = [upcase %wit%]
&end

/* Creating point coverage from houses polygon coverage if necessary
&if not [null %rha%] &then
    &do
        &sv odp = [show display]
        display 0
        arccedit
        edit %rha%
        editfeature label
        select all
        put %rhp%
        quit
        display %odp%
        &if %wis% = SC &then
            alter AREA %wis%,,,,,
        build %rhp% points
    &end

/* Overlay coverages
identity %rhp% %cen% %rho% point 0.5 join
dropitem %rho%.PAT %rho%.PAT %rhpen%# %rhpen%-ID %cenen%# %cenen%-ID PR

/* Summarize built-up surfaces
frequency %rho%.PAT %cen%.RHO
CTR-ID
END
&if %wis% <> N &then
    %wit%
END
&sv ows = [show workspace]
workspace [dir %rho%]
tables
dropitem %cen%.RHO CASE#
select %cen%.RHO
alter FREQUENCY N,,,,,
select

/* Calculate weighted population of each building
```

```

additem %rho%.PAT PR 8 12 B
relate add
CSEZ %cen%.PAT info CTR-ID CTR-ID linear ro
FREQ %cen%.RHO info CTR-ID CTR-ID linear ro
[unquote '']
select %rho%.PAT
calculate PR = int(((%wit% * CSEZ//PR) / FREQ//%wis%) + 0.5)
relate drop CSEZ FREQ [unquote '']
quit
workspace %odp%

/* Re-create houses polygon coverage if necessary
&if not [null %rha%] &then
  &do
    display 0
    arccedit
    edit %rho%
    editfeature lines
    weedtolerance 0
    intersectarcs off
    nodesnap closest 0
    get %rha%
    save
    quit
    display %odp%
    build %rho% nodes
    build %rho% lines
    build %rho%
  &end

```

3 Modello di Integrazione ‘Fuzzy’ dell’Uso del Suolo (FLCIM)

Il modello di integrazione ‘fuzzy’ dell’uso del suolo (*Fuzzy Land-Cover Integration Model*, FLCIM) si propone di fornire una soluzione soprattutto al problema dell’integrazione di modelli numerici sull’uso del suolo incompleti da un punto di vista tematico a mezzo di dati con errori posizionali maggiori provenienti prevalentemente da rilievi satellitari. Sebbene le tecniche geo-informatiche — a differenza di quelle cartografiche tradizionali — rendano possibili integrazioni strutturali tra dati aventi caratteristiche qualitative (errori posizionali, accuratezza o ‘scala’) diverse, da un punto di vista teorico tali operazioni sono tuttavia altamente sconsigliabili in quanto viene sensibilmente alterata la significatività sia delle informazioni in ‘entrata’ che del dato finale in ‘uscita’. Per ovviare all’introduzione di questo tipo di incoerenza geometrica, il FLCIM integra due insiemi di dati con impianto bidimensionale (topologia poligonale) lasciando inalterate le informazioni metriche del dato con accuratezza maggiore aggiun-

gendovi un livello di informazioni tematiche provenienti dal dato con accuratezza minore.

Principio

Il principio guida della procedura è quello del calcolo della *probabilità che una porzione di territorio appartenga ad un determinato tematismo*. Supponendo ad esempio di disporre di una copertura poligonale «uso del suolo» con un errore posizionale medio attorno a 0,50 m e di un archivio basato su celle (raster) estratto da un rilievo satellitare multispettrale «boschi ad alto fusto» con un errore posizionale medio attorno a 10,00 m, il compito da svolgere si presenta, da un punto di vista concettuale, nel modo seguente.

1. Occorre trasformare il valore cromatico assegnato ad ogni cella che indica l'estensione dei boschi ad alto fusto in valori percentuali di probabilità che una determinata cella sia da considerare come appartenente ad un bosco ad alto fusto. Soprattutto nelle zone marginali dei boschi, tali valori percentuali saranno — con gradazioni differenti — inferiori a 100 % e potranno avvicinarsi a 0 % nelle zone più esterne; nel caso di boschi con estensioni molto ridotte (con larghezze inferiori al doppio dell'errore posizionale medio) potranno essere completamente assenti celle con valori percentuali uguali a 100 %.
2. Convertendo il dato basato su celle in copertura poligonale, si ottiene un insieme di poligoni con valori percentuali omogenee che potrà essere integrato nell'uso del suolo con un'operazione di sovrapposizione «polygon-in-polygon» mantenendo gli attributi poligonali di entrambe le coperture.
3. Il valore percentuale medio per ogni poligono originale dell'uso del suolo sarà dato da

$$P = \sum_{i=1}^n \frac{A_i P_i}{A}, \quad (5)$$

dove n è il numero di poligoni della copertura relativa ai boschi ad alto fusto che cadono all'interno di un singolo poligono dell'uso del suolo, A la superficie di un singolo poligono dell'uso del suolo, A_i la superficie di un singolo poligono boschivo dopo l'operazione di sovrapposizione con l'uso del suolo e P_i il valore percentuale di un singolo poligono boschivo.

4. I valori percentuali medi riferiti ai singoli poligoni dell'uso del suolo potranno infine essere associati al dato originale dell'uso del suolo. Il dato temporaneo prodotto tramite l'operazione di sovrapposizione — e che presenterà delle alterazioni geometriche di entrambi i dati — può a questo punto essere eliminato.

Implementazione

Un'implementazione del FLCIM richiede che siano disponibili due insiemi di dati geografici.

Tabella 3: Dati territoriali di base richiesti per il modello di integrazione ‘fuzzy’ dell’uso del suolo (FLCIM)

Dato	Entità	Attributi	Note
Uso del suolo	poligoni		richiesto
Tematismo integrativo	raster o poligoni	campi numerici	richiesto

1. Una *copertura poligonale* che codifica una suddivisione (mosaico) in porzioni (lotti, aree di pertinenza o altri ‘atomi’ territoriali) del territorio preso in esame (vedi «Copertura originale» in Figura 1). A tali porzioni di territorio dovrà essere aggiunto un nuovo attributo che documenta la probabilità dell’appartenenza tematica all’insieme integrativo.
2. Una *copertura poligonale* con un campo numerico intero o un *dato geospaziale basato su celle* (raster) codificato con numeri interi che descrive l’intensità del tematismo integrativo (vedi «Dato integrativo» in Figura 1). I valori numerici possono contenere sia valori gradualmente che valori booleani (per esempio 0 e 1).

Le operazioni che, da un punto di vista geo-informativo, devono essere eseguite sono le seguenti.

1. *Standardizzazione dei valori numerici*. I valori numerici interi presenti nel dato integrativo non devono necessariamente essere già codificati secondo l’intervallo convenzionale $[0; 100]$, ma possono essere compresi entro un intervallo $[a; b]$ generico. La percentuale standardizzata del valore di una singola cella o di un singolo poligono viene quindi calcolata con


$$P_i = \frac{100(V_i - a)}{b - a} \quad (6)$$

dove a e b sono i valori minimo e massimo dell’intervallo $[a; b]$ e V_i il valore non standardizzato della cella o del poligono. La standardizzazione avviene, nel caso di dati basati su celle, impiegando le tecniche della *map-algebra* oppure, per le coperture poligonali, predisponendo un nuovo campo numerico e calcolando i suoi valori secondo la formula esposta.

2. Nel caso il dato integrativo sia basato su celle, sarà necessario effettuare la *conversione in copertura poligonale* per poter eseguire le consuete operazioni di sovrapposizione «polygon-in-polygon». Tale conversione non è da confondere con il processo di vettorializzazione di un dato raster, in quanto le celle — entità quadratiche da un punto di vista geometrico — vengono semplicemente convertite in poligoni regolari delimitati da quattro linee di bordo e quattro nodi.
3. *Sovrapposizione poligonale dei due insiemi di dati*. L’unione delle due coperture viene effettuata con il comando `union` che preserva tutti gli attributi poligonali delle coperture di origine e consente di determinare quali porzioni — comprese le relative informazioni metriche — di quali

3 Modello di Integrazione 'Fuzzy' dell'Uso del Suolo (FLCIM)

Figura 1: Schema procedurale delle operazioni di analisi spaziale del FLCIM



celle o conglomerati di celle del dato integrativo ricadono all'interno di quali poligoni della copertura poligonale originale. Si potrà così ottenere una tabella di corrispondenza tra i due insiemi di dati nella quale è riportata, per ogni poligono della copertura originale, una serie di records con la coppia di attributi *area* (superficie) e *value* (valore percentuale) (vedi «Sovrapposizione» in Figura 1).

4. *Calcolo delle somme ponderate dei valori percentuali* per ogni singolo poligono della copertura di unione. Alla coppia di attributi *area* e *value* va aggiunto un terzo campo

$$\Sigma(value) = area \cdot value \quad (7)$$

che — nonostante appaia in questa equazione come prodotto — è in realtà la somma aritmetica dei valori percentuali delle singole celle (aventi ciascuna una superficie uguale) del dato di sovrapposizione.

5. *Calcolo delle somme ponderate dei valori percentuali* per ogni singolo poligono della copertura originale. A partire dalla tabella degli attributi poligonali della copertura di unione occorre ora costruire una tabella di riepilogo che riporti, per ciascun poligono della copertura originale, le somme delle somme $\Sigma(value)$ calcolate durante il passaggio precedente. Supponendo che la copertura di origine si chiami LANDUSE e il nome del campo numerico che ospita i valori percentuali FOREST, si dovranno calcolare le somme del campo S.FOREST per i records con lo stesso identificatore poligonale LANDUSE# nel modo illustrato con la seguente sequenza di istruzioni.

```
frequency landuse.pat landuse.forest
forest#
end
s_forest
end
```

6. *Creazione della copertura integrata.* Dato che non è consigliabile alterare direttamente la copertura originale, ma aggiungere il livello tematico integrativo ad una nuova copertura (sarà poi l'utente a decidere se mantenere la versione iniziale o meno). Supponendo che tale copertura si chiami LANDUSE2, verrà creata una copia della copertura originale con l'istruzione `copy landuse landuse2` e aggiunto il campo numerico che ospiterà i valori percentuali definitivi.
7. *Impostazione di un ambiente di relazionamento temporaneo.* Per procedere e calcolare i valori percentuali riferiti alla media nell'ambito dei singoli poligoni della copertura originale, è necessario impostare un ambiente di relazionamento verso la tabella di riepilogo creata nel passaggio precedente utilizzando come campo chiave nella tabella di riepilogo il campo LANDUSE# e nella copertura da integrare il campo LANDUSE2#. Dato che, fino a questo punto, le due coperture LANDUSE e LANDUSE2 sono copie esatte, anche gli identificatori poligonali delle due coperture sono equivalenti.
8. *Calcolo dei valori percentuali* per ogni singolo poligono della copertura integrata. Le operazioni di calcolo per ottenere la somma $\Sigma(value)$ co-

dificata nel campo S_FOREST della tabella di riepilogo corrisponde — per riprendere la forma sintattica utilizzata in precedenza — all'espressione

$$S = \sum_{i=1}^n A_i P_i \quad (8)$$

che — se inserita nell'Equazione (5) — semplifica l'algoritmo centrale in

$$P = \frac{S}{A} \quad (9)$$

Ritornando ora ai nomi dei campi assunti qui a titolo d'esempio, quest'ultima equazione può essere tradotta nell'istruzione finale
`calculate forest = valsums//s_forest / area`
 dove VALSUMS è il nome della relazione precedentemente impostata. Dopodiché possono essere eliminati l'ambiente di relazionamento VALSUMS e la copertura temporanea XXTEMP.

Procedura

La procedura riportata qui di seguito rappresenta un'implementazione completa del FLCIM per Arc/Info sotto forma di un macro-comando con la sintassi

```
FLACIM <land_cover> <integration_dataset> <out_cover> <integer_item>
      {min} {max} {fuzzy_tolerance}
```

dove

`land_cover` indica la copertura poligonale contenente un modello numerico degli usi del suolo,

`integration_dataset` il modello basato su celle o la copertura poligonale contenente i dati con cui integrare il modello degli usi del suolo,

`out_cover` la copertura poligonale da creare,

`integer_item` il campo numerici dell'insieme di dati integrativo,

`min` il numero intero che indica una probabilità dello 0% che il poligono o la cella in questione appartenga al tematismo codificato dal campo `<item>` (il valore di default è 0),

`max` il numero intero che indica una probabilità del 100% che il poligono o la cella in questione appartenga al tematismo codificato dal campo `<item>` (il valore di default è 100) e

`fuzzy_tolerance` la tolleranza (errore posizionale medio ammesso) della copertura finale.

```
&args lco sco oco itm min max fto
```

```
/* Parameter tests
```

```
&if [null %itm%] & then
```

```
&do
```

```
&type Usage: FLCIM <land_cover> <integration_dataset> <out_cover>
```

```
&type <integer_item> {min} {max} {fuzzy_tolerance}
```

```

    &return
    &end
    &if [null %min%] &then min = 0
    &if [null %max%] &then max = 100
    &if [null %fto%] &then &sv fto
    &sv lco = [upcase %lco%]
    &sv sco = [upcase %sco%]
    &sv oco = [upcase %oco%]
    &sv itm = [upcase %itm%]
    &sv ows = [show workspace]
    &sv odp = [show display]
    workspace [dir %sco%]
    display 0

    &if [exists %sco% -polygon &then]

    /* Handle polygon coverage if appropriated
    &do
    copy %sco% xtemp
    tables
    additem xtemp.pat GRID-CODE 4 5 b
    select xtemp.pat
    calculate GRID-CODE = (100 * (%itm% - %min%)) / (%max% - %min%)
    quit
    &end

    &else &if [exists %sco% -grid] &then

    /* Handle grid if appropriated
    &do
    grid
    xtempg = (100 * (%sco% - %min%)) / (%max% - %min%)
    quit
    gridpoly xtempg xtemp 0.0
    kill xtempg all
    &end

    /* Otherwise exit
    &else &return Warning: Dataset %sco% is neither a polygon coverage nor a grid.

    /* Perform polygon overlay and calculate weighted values
    union %ico% xtemp xtempu %fto%
    kill xtemp all
    rename xtempu xtemp
    tables
    additem xtemp.pat S_%itm% 4 5 b
    select xtemp.pat
    calculate S_%itm% = AREA * GRID-CODE
    quit

```

3 Modello di Integrazione 'Fuzzy' dell'Uso del Suolo (FLCIM)

```
/* Create output coverage and calculate percent values
copy %lco% %oco%
frequency xtemp.pat xtemp.frq
[unquote [entryname %lco%]]#
end
S_[unquote %itm%]
end
tables
relate add
VALSUMS xtemp.frq info
[unquote [entryname %lco%]]# [unquote [entryname %lco%]]#
linear ro
[unquote '']
additem [entryname %oco%].pat P_%itm% 4 5 b
select [entryname %oco%].pat
/* This is the central formula
calculate P_%itm% = VALSUMS//S_[unquote %itm%]) / AREA
relate drop VALSUMS [unquote '']
quit
kill xtemp all

/* Reset previous environment variables
workspace %ows%
display %odp%
```